From Hilbert to Tarski<br>Gabriel Braun, Pierre Boutry, Julien Narboux

## To cite this version:

Gabriel Braun, Pierre Boutry, Julien Narboux. From Hilbert to Tarski. Eleventh International Workshop on Automated Deduction in Geometry, Jun 2016, Strasbourg, France. pp.19. hal-01332044

HAL Id: hal-01332044
https://inria.hal.science/hal-01332044
Submitted on 15 Jun 2016

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire HAL, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

# From Hilbert to Tarski 

Gabriel Braun, Pierre Boutry, and Julien Narboux<br>ICube, UMR 7357 CNRS, University of Strasbourg<br>Pôle API, Bd Sébastien Brant, BP 10413, 67412 Illkirch, France<br>\{gabriel.braun, boutry, narboux\}@unistra.fr


#### Abstract

In this paper, we describe the formal proof using the Coq proof assistant that Tarski's axioms for plane neutral geometry (excluding continuity axioms) can be derived from the corresponding Hilbert's axioms. Previously, we mechanized the proof that Tarski's version of the parallel postulate is equivalent to the Playfair's postulate used by Hilbert [9] and that Hilbert's axioms for plane neutral geometry (excluding continuity) can be derived from the corresponding Tarski's axioms [12]. Hence, this work allows us to complete the formal proof of the equivalence between the two axiom systems for neutral and plane Euclidean geometry. Formalizing Hilbert's axioms is not completely straightforward, in this paper we describe the corrections we had to make to our previous formalization. We mechanized the proof of Hilbert's theorems that are required in our proof of Tarski's axioms. But this connection from Hilbert's axioms, allows to recover the many results we obtained previously in the context of Tarski's geometry: this includes the theorems of Pappus[13], Desargues, Pythagoras and the arithmetization of geometry [8].


Keywords: formal proof, geometry, foundations, Tarski's axioms, Hilbert's axioms, Coq

## 1 Introduction

There are several ways to define the foundations of geometry. In the synthetic approach, the axiomatic system is based on some geometric objects and axioms about them. The best known modern axiomatic systems based on this approach are Hilbert's [17] and Tarski's ones [32]. In the analytic approach, a field $\mathbb{F}$ is assumed (usually $\mathbb{R}$ ) and the space is defined as $\mathbb{F}^{n}$. In the mixed analytic/synthetic approaches, one assumes both the existence of a field and also some geometric axioms. For example, the axiomatic systems for geometry used for education in North America are based on Birkhoff's axiomatic system [7] in which the field serves to measure distances and angles. This is called the metric approach. A modern development of geometry based on this approach can be found in the books of Millman [21] or Moise [22]. The textbook of Hartshorne [16] provides a development of geometry based on Hilbert's axioms. A rigorous development of geometry based on Tarski's axioms appears in the first part of the book by Tarski, Szmielew and Schwabhäuser [27].

The GeoCoq library gathers results about the foundations of geometry formalized in Coq [23]. It includes a formalization of the first part of [27]. We formalized in Coq the
link between the synthetic geometry defined by Tarski's axioms and analytic geometry [8]. We also formalized the link from Tarski's axioms to Hilbert's axioms [12], Beeson has later written a note [5] to demonstrate that the main results to obtain Hilbert"s axioms are contained in [27]. In this paper, we complete the picture, by proving formally that Tarski's axioms can be derived from Hilbert's axiom.

Dehlinger, Dufourd and Schreck have studied the formalization of Hilbert's foundations of geometry in the intuitionist setting of Coq [14]. They have highlighted the fact that Hilbert's proofs require the excluded middle axiom applied to the point equality and the role of non degenerate conditions. They focus on the first two groups of axioms and prove some betweenness properties. Meikle and Fleuriot have done a similar study within the Isabelle/HOL proof assistant [20]. They went up to twelfth ${ }^{1}$ theorem of Hilbert's book. Scott has continued the formalization of Meikle using Isabelle/HOL and revised it [28]. He has corrected some "subtle errors in the formalization of Group III by Meikle". Scott was interested in trying to obtain readable proofs. Later, he developed a system within the HOL-Light proof assistant to automatically fill some gaps in the incidence proofs [30]. In our work, even if we use some tactics to obtain some proofs more easily [11,10], we do not focus on obtaining readable proofs. Scott then focused on proving a theorem, which Hilbert's states without proof: that a special case of the Jordan Curve Theorem can be proved using the first two groups of Hilbert's axioms only [29]. Moreover Richter has formalized a substantial number of results based on Hilbert's axioms and a metric axiom system using HOL-Light [25]. Finally Beeson and Wos have used the OTTER automatic theorem prover to mechanize some proofs in Tarski's geometry [3,4]. They manage to automatically prove many results from the first twelve chapter of [27], with the help of some human generated hints OTTER can prove all the lemmas necessary to derive Hilbert's axioms.

The earlier errors in Meikle's formalization found by Scott can frighten the reader. How can we be sure that there are no errors in the formalization presented in this paper? They are two kinds of potential errors:

1. the axioms are inconsistent
2. the axioms do not capture the desired theory - some standard geometry theorems may not be provable from the axioms

In our previous work [12], we have shown that Tarski's axioms for plan Euclidean geometry can serve as a model for the corresponding Hilbert's axioms. In an unpublished result, Boutry has built a formal proof that the Cartesian plane over a Pythagorean ordered field is a model of Tarski's axioms (A1-A10, i.e. excluding the continuity axiom). This can convince the reader that Tarski's axioms as they are formalized are consistent and hence our formalization of Hilbert's axiom system as well. Makarios has also shown using the Isabelle/HOL proof assistant that there are both euclidean and a non-euclidean models of Tarski's axioms for plane neutral geometry [19]. However our axiom system could be weaker than necessary. As a matter of fact, the axiom system we proposed in 2012 was not sufficient to prove Tarski's axioms and we had to correct it. In this paper, we present a formal proof that the formalization of Hilbert's axioms is not only correct

[^0]but also sufficient, in the sense that we can obtain the culminating result of both [17] and [27]: the arithmetization of geometry.

## Outline

The paper is organized as follows. In Section 2, we describe Tarski's axiom system (the version used in [27]), then (Sec. 3) our formalization of Hilbert's axioms for plane Euclidean geometry. In doing so, we present the errors we had to correct in our previous axiomatic system to obtain the equivalence with Tarski's axioms. Finally, in Section 4, we provide the proof that Tarski's axioms can be derived from Hilbert's axioms.

## 2 Tarski's axioms

We use the axioms that serves as a basis for [27]. For an explanation of the axioms and their history see [32]. Table 1 lists the axioms for neutral geometry.

Let us recall that Tarski's axiom system is based on a single primitive type depicting points and two predicates, namely between noted by -- and congruence noted by $\equiv$. $A-B-C$ means that $A, B$ and $C$ are collinear and $B$ is between $A$ and $C$ (and $B$ may be equal to $A$ or $C$ ). $A B \equiv C D$ means that the line-segments $\overline{A B}$ and $\overline{C D}$ have the same length (the orientation does not need to be the same).

Figure 2 provides the Coq's formalization of these axioms. We worked in an intuitionistic setting but we included the axiom point_equality_decidability.

$$
\begin{array}{lc}
\hline \text { A1 } & \text { Symmetry } A B \equiv B A \\
\text { A2 } & \text { Pseudo-Transitivity } A B \equiv C D \wedge A B \equiv E F \Rightarrow C D \equiv E F \\
\text { A3 } & \text { Cong Identity } A B \equiv C C \Rightarrow A=B \\
\text { A4 } & \text { Segment construction } \exists E, A-B-E \wedge B E \equiv C D \\
\text { A5 } & \text { Five-segment } A B \equiv A^{\prime} B^{\prime} \wedge B C \equiv B^{\prime} C^{\prime} \wedge \\
& A D \equiv A^{\prime} D^{\prime} \wedge B D \equiv B^{\prime} D^{\prime} \wedge \\
& A-B-C \wedge A^{\prime}-B^{\prime}-C^{\prime} \wedge A \neq B \Rightarrow C D \equiv C^{\prime} D^{\prime} \\
\text { A6 } & \text { Between Identity } A-B-A \Rightarrow A=B \\
\text { A7 } & \text { Inner Pasch } A-P-C \wedge B-Q-C \Rightarrow \exists X, P-X-B \wedge Q-X-A \\
\text { A8 } & \text { Lower Dimension } \exists A B C, \neg A-B-C \wedge \neg B-C-A \wedge \neg C-A-B \\
\text { A9 } & \text { Upper Dimension } A P \equiv A Q \wedge B P \equiv B Q \wedge C P \equiv C Q \wedge P \neq Q \\
& \Rightarrow A-B-C \vee B-C-A \vee C-A-B . \\
\text { A10 } & \text { Parallel postulate } \exists X Y(A-D-T \wedge B-D-C \wedge A \neq D \Rightarrow \\
& A-B-X \wedge A-C-Y \wedge X-T-Y) \\
\hline
\end{array}
$$

Table 1: Tarski's axiom system for neutral geometry.


Fig. 1: Illustration for three axioms

The symmetry axiom (A1 on Table 1) for equidistance together with the transitivity axiom (A2) for equidistance imply that the equi-distance relation is an equivalence relation. The identity axiom for equidistance (A3) ensures that only degenerate line segments can be congruent to a degenerate line segment. The axiom of segment construction (A4) allows to extend a line segment by a given length. The five-segments axiom (A5) is similar to the Side-Angle-Side principle, but expressed without mentioning angles, using only the betweenness and congruence relations (Fig. 1a). The lengths of $\overline{A B}, \overline{A D}$ and $\overline{B D}$ fix the angle $\widehat{C B D}$. The identity axiom for betweenness expresses that the only possibility to have $B$ between $A$ and $A$ is to have $A$ and $B$ equal. Tarski's relation of betweenness is non-strict, unlike Hilbert's. The inner form of the Pasch's axiom (A7, Fig. 1b) is a variant of the axiom that Pasch introduced in [24] to repair the defects of Euclid. Pasch's axiom intuitively says that if a line meets one side of a triangle and does not pass through the endpoints of that side, then it has to meet one of the other sides of the triangle. Inner Pasch's axiom is a form of the axiom that holds even in 3 -space, i.e. does not assume a dimension axiom. The lower 2 -dimensional axiom (A8) asserts that the existence of three non-collinear points. The upper 2-dimensional axiom (A9) implies that all the points are coplanar. The version of the parallel postulate (A10) is a statement which can be expressed easily in the language of Tarski's geometry (Fig. 1c). It is equivalent to the uniqueness of parallels or Euclid's 5th postulate.

```
Class Tarski_neutral_dimensionless :=
{
    Tpoint : Type;
    Bet : Tpoint -> Tpoint -> Tpoint -> Prop;
    Cong : Tpoint -> Tpoint -> Tpoint -> Tpoint -> Prop;
    cong_pseudo_reflexivity : forall A B, Cong A B B A;
    cong_inner_transitivity : forall A B C D E F,
        Cong A B C D -> Cong A B E F -> Cong C D E F;
    cong_identity : forall A B C, Cong A B C C }->\mathrm{ A A = B;
    segment_construction : forall A B C D,
        exists E, Bet A B E /\ Cong B E C D;
    five_segment : forall A A' B B' C C' D D',
        Cong A B A' B' ->
        Cong B C B' C' ->
        Cong A D A' D' ->
        Cong B D B' D' ->
        Bet A B C -> Bet A' B' C' -> A <> B -> Cong C D C' D';
    between_identity : forall A B, Bet A B A >> A = B;
    inner_pasch : forall A B C P Q,
        Bet A P C -> Bet B Q C ->
        exists X, Bet P X B /\ Bet Q X A;
    PA : Tpoint;
    PB : Tpoint;
    PC : Tpoint;
    lower_dim : ~ (Bet PA PB PC \/ Bet PB PC PA \/ Bet PC PA PB)
}.
Class Tarski_neutral_dimensionless_with_decidable_point_equality
    `(Tn : Tarski_neutral_dimensionless) :=
{
    point_equality_decidability : forall A B : Tpoint, A = B \/ ~ A = B
}.
Class Tarski_2D
    `(TnEQD : Tarski_neutral_dimensionless_with__decidable_point_equality) :=
{
    upper_dim : forall A B C P Q,
        P <> Q -> Cong A P A Q -> Cong B P B Q -> Cong C P C Q ->
        (Bet A B C \/ Bet B C A \/ Bet C A B)
}.
Class Tarski_2D_euclidean '(T2D : Tarski_2D) :=
{
    euclid : forall A B C D T,
        Bet A D T -> Bet B D C -> A<>D ->
        exists X, exists Y,
        Bet A B X /\ Bet A C Y /\ Bet X T Y
}.
```

Fig. 2: The formalization of Tarski's axioms

## 3 Hilbert's axioms

Our formalization of Hilbert's axiom system is derived on the French translation of the tenth edition annotated by Rossier [18].

Hilbert's axiom system is based on two abstract types: points and lines (as we limit ourselves to 2-dimensional geometry we did not introduce 'planes' and the related axioms).

In our initial version of Hilbert's axioms, we made several mistakes. None of the axioms were incorrect (as they are formally proved from Tarski's axioms), but some should be strengthened and some others are useless because they can be derived. For each group of axioms, we detail the changes we made to our previous formalization, we do not detail all the axioms that have already been described in [12]. But, we give the full list of axioms as a reference.

### 3.1 Group I

Group I of axioms contains the incidence axioms.
First, we had to change the lower dimensional axiom. Hilbert states that there exists three non collinear points and three points are said to be collinear if there exists a line going through these three points. This assumption is not enough, because in a world without lines, assuming that there are three non collinear points does not imply that they are distinct. Indeed, there is a model of the first two groups of Hilbert's axioms with only one point and no lines (interpreting congruence by the empty relation). We can construct a line only if we have two distinct points. Scott's formalization does not need this modification because in Isabelle/HOL all types are inhabited, hence his formalization includes implicitly the fact that there is at least one line. Meikle's and Richter's formalizations enforce that the three points are distinct.

Hence, for the lower dimension axiom, we state that there is a line and point not on this line:

```
10 : Line;
P0 : Point;
plan : ~ Incid PO l0;
```

As there is an axiom stating there are always at least two points on a line, the former axiom stating that there are three non collinear points can be derived.

Second, we had to introduce the property that line equality is an equivalence relation and that incidence is a morphism for line equality:

```
EqL_Equiv : Equivalence EqL;
Incid_morphism :
    forall P l m, Incid P l -> EqL l m -> Incid P m;
```

Finally, as we are working in an intuitionist setting we had to introduce some decidability properties which allow to perform case distinctions. It would be interesting to formalize a constructive version of Hilbert's axioms, following Beeson's work [1], we leave this for future work.

```
Incid_dec : forall P l, Incid P l \/ ~Incid P l;
eq_dec_pointsH : forall A B : Point, A=B \/ ~ A=B;
```

The complete list of axioms for group I is given in Fig. 3.

```
Point : Type;
Line : Type;
EqL : Line -> Line -> Prop;
EqL_Equiv : Equivalence EqL;
Incid : Point -> Line -> Prop;
Incid_morphism :
    forall P l m, Incid P l -> EqL l m -> Incid P m;
Incid_dec : forall P l, Incid P l \/ ~ Incid P l;
eq_dec_pointsH : forall A B : Point, A=B \/ ~ A=B;
line_existence :
    forall A B, A <> B -> exists l, Incid A l /\ Incid B l;
line_uniqueness :
    forall A B l m,
        A <> B ->
        Incid A l -> Incid B l -> Incid A m -> Incid B m ->
            EqL l m;
two_points_on_line :
    forall l,
        { A : Point & { B | Incid B l /\ Incid A l /\ A <> B}};
ColH :=
    fun A B C => exists l, Incid A l /\ Incid B l /\ Incid C l;
l0 : Line;
PO : Point;
plan : ~ Incid PO 10;
```

Fig. 3: Formalization of Group I

### 3.2 Group II

Group II of axioms contains the betweenness axioms. We denote by $A \leadsto B \leadsto C$ Hilbert's betweenness predicate, which is strict. It expresses the fact that $B$ is on the line $A C$ between $A$ and $C$ and different from $A$ and $C$. We could not derive the fact that if $A \rightarrow B \rightarrow C$ then $A \neq C$ from our former axioms so we added this property as an axiom. The fact that $A \neq B$ and $B \neq C$ (which is assumed by Greenberg, Hartshorne and Richter) is not necessary as it can be derived from the other axioms. The fact that $A$ should be different from $C$ is not explicit in Hilbert's book.

The property between_one states that given three collinear points distinct points at least one of them is between the other two:

```
between_one :
    forall A B C,
        A <> B -> A <> C -> B <> C -> ColH A B C ->
        BetH A B C \/ BetH B C A \/ BetH B A C.
```

In our earlier formalization as well as earlier editions of Hilbert's book, this property was taken as an axiom. Following the proof by Wald published by Hilbert in later editions, we derived it from the other axioms. Richter assumes this property. Moreover, in the axiom between_only_one, we removed one of the conjuncts as it can be derived. We now have:
between_only_one : forall A B C, BetH A B C $->$ ~ BetH B C A;
instead of:

```
between_only_one :
forall A B C, BetH A B C -> ~ BetH B C A /\ ~ BetH B A C;
```

The other axioms could be kept unmodified. The axioms for the second group are given in Fig. 4.

```
BetH : Point -> Point -> Point -> Prop;
between_col : forall A B C, BetH A B C -> ColH A B C;
between_diff : forall A B C, BetH A B C -> A <> C;
between_comm : forall A B C, BetH A B C >> BetH C B A;
between_out : forall A B, A <> B -> exists C, BetH A B C;
between_only_one : forall A B C, BetH A B C -> ~ BetH B C A;
cut :=
    fun l A B => ~ Incid A l /\ ~ Incid B l 八
        exists I, Incid I I /\ BetH A I B;
pasch :
    forall A B C l,
        ~ ColH A B C -> ~ Incid C l -> cut l A B ->
        cut l A C \/ cut l B C;
```

Fig. 4: Formalization of Group II

### 3.3 Group III

Group III of axioms contain those about congruence of segments and angles.

Congruence of segments Hilbert defines congruence as a relation about segments, where segments are defined as unordered pairs of points. In our formalization, we chose to avoid defining the concept of segment. We denote by $\equiv_{H}$ the Hilbert segment congruence relation. Hence, we have an axiom which says that segments can be permuted on the right. Other permutations can be derived.

```
cong_permr : forall A B C D, CongH A B C D -> CongH A B D C;
```

The uniqueness of segment construction can be derived if one assume the reflexivity of congruence of angles, therefore we dropped this axiom. Richter assumes uniqueness of segment construction. The other axioms were not changed, the full list is given on Fig. 5. Our formalization of the segment addition axiom follows Hilbert's prose. It is based on the definition of the concept of disjoint segments. Note that in the segment addition axiom, the concept of disjoint segments could be replaced by a betweenness assumption stating that $B$ is between $A$ and $C$ and $B^{\prime}$ is between $A^{\prime}$ and $C^{\prime}$, we proved it as lemma:

```
Lemma addition_betH : forall A B C A' B' C',
    BetH A B C -> BetH A' B' C' ->
    CongH A B A' B' -> CongH B C B' C' ->
    CongH A C A' C'.
```

Congruence of angles In early editions of the Foundations of Geometry, Hilbert had taken pseudo-transitivity of congruence of angles as an axiom. Later Rosenthal has shown that this axiom can be derived from the others [26]. We used the later version of Hilbert's axioms. Note that, as we need transitivity of congruence in our proofs, we had to formalize Rosenthal's proofs. These proofs are rather technical, not very interesting in a pedagogical context, that is why Hartshorne chose to take the simpler, not minimal, axiom system of the earlier editions [16]. Richter also assumes the transitivity of congruence of angles. In our previous formalization, we defined the concept of angles $A B C$ as three points $A, B$ and $C$, with a proof that $A \neq B$ and $B \neq C$. We did not want to open the Pandora box of the dependent types [6] (types that depend on a proof), so we dropped the proofs from the angle type, and even made implicit the angle type. To be faithful to Hilbert, some non degeneracy conditions are added to ensure that angles are neither flat nor null. This makes the proof of the Tarski's five-segment axiom (A5) more involved.

We used a predicate of arity six for the congruence of angles:

```
CongaH :
    Point -> Point -> Point -> Point -> Point -> Point -> Prop;
```

```
CongH : Point -> Point -> Point -> Point -> Prop;
cong_permr : forall A B C D, CongH A B C D -> CongH A B D C;
cong_pseudo_transitivity :
    forall A B C D E F,
        CongH A B C D -> CongH A B E F -> CongH C D E F;
cong_existence :
    forall A B l M,
        A <> B -> Incid M l ->
        exists A', exists B', Incid A' l /\ Incid B' l /\
                            BetH A' M B' /\
                                    CongH M A' A B /\ CongH M B' A B;
disjoint := fun A B C D => ~ exists P, BetH A P B /\ BetH C P D;
addition :
    forall A B C A' B' C',
        ColH A B C -> ColH A' B' C' ->
        disjoint A B B C -> disjoint A' B' B' C' ->
        CongH A B A' B' -> CongH B C B' C' ->
        CongH A C A' C';
```

Fig. 5: Formalization of Group III, part 1 : segment congruence axioms

As for the congruence of segments we need a permutation property about angle congruence:

```
congaH_permlr :
    forall A B C D E F, CongaH A B C D E F -> CongaH C B A F E D;
```

Our approach does not use rays, so we need to state that two angles represented by the same rays are congruent. This is the purpose of axiom congaH_outh_congaH. The predicate outh P A B states that $B$ belongs to the ray $P A$ :

```
outH :=
    fun P A B => BetH P A B \/ BetH P B A \/ (P <> A 八\A = B);
congaH_outH_congaH :
    forall A B C D E F A' C' D' F',
        CongaH A B C D E F ->
        outH B A A' -> outH B C C' -> outH E D D' -> outH E F F' ->
        CongaH A' B C' D' E F';
```

Recall that Hilbert's axiom III 4 states that:
Given an angle $\alpha$, a ray $h$ emanating from a point $O$ and given a point $P$, not on the line generated by $h$, there is a unique ray $h^{\prime}$ emanating from $O$, such that the angle $\alpha^{\prime}$ defined by $\left(h, O, h^{\prime}\right)$ is congruent with $\alpha$ and such that every point inside $\alpha^{\prime}$ and $P$ are on the same side with respect to the line generated by $h$.

We simplified the formalization of Hilbert's axiom III 4, instead of considering every point inside the angle, our proof shows that it is sufficient to state that the point that defines the angle is on the same side as $P$. Hence, we can save the burden of defining what it means for a point to be inside an angle. Our version is also simpler than Scoot's one, which follows Hilbert's definition.

We say that two points are on the same side of a line, if there is a point $P$ such that they are both on opposite sides wrt. $P$. The fact that two points are on opposite sides of a line is defined by the cut predicate of Group II.

```
hcong_4_existence :
    forall A B C O X P,
        ~ ColH P O X -> ~ ColH A B C ->
        exists Y, CongaH A B C X O Y /\ same_side' P Y O X;
hcong_4_uniqueness :
    forall A B C O P X Y Y',
        ~ ColH P O X -> ~ ColH A B C ->
        CongaH A B C X O Y -> CongaH A B C X O Y' ->
        same_side' P Y O X -> same_side' P Y' O X ->
        outH O Y Y'
```

The full list of axioms is given in Fig. 6.

### 3.4 Group IV

In 2012, we had an axiom saying that given a line and a point there exists a parallel line through this point. Only the uniqueness of the parallel should be assumed as the existence can be derived from other axioms. The uniqueness axiom is depicted on Fig. 7.

## 4 Proving Tarski's axioms

To prove Tarski's axioms we use a trick: we worked on both sides. We derived some lemmas in the context of Hilbert's axioms and we also proved Tarski's axioms in the context of a variant of these axioms. We introduced this variant of Tarski's axiom system inspired by the one adopted by Beeson in [2]. We added the following two properties as axioms: symmetry of betweenness and inner transitivity and we number them as in [32]. We modified Pasch's axiom to have a version which excludes the degenerate case when the triangle $A B C$ is flat. This non degeneracy condition is crucial because obtaining the general case from Hilbert's version is not trivial.

Formally, we consider the following axioms:

$$
\begin{aligned}
A_{14}:= & \forall A B C, A-B-C \Rightarrow C-B-A \\
A_{15}:= & \forall A B C D, A-B-D \wedge B-C-D \Rightarrow A-B-C \\
A_{7}^{\prime}:= & \forall A B C P Q, A-P-C \wedge B-Q-C \wedge \\
& A \neq P \wedge P \neq C \wedge B \neq Q \wedge Q \neq C \wedge \neg \operatorname{Col}(A B C) \Rightarrow \\
& \exists X, P-X-B \wedge Q-X-A
\end{aligned}
$$

```
CongaH :
    Point -> Point -> Point -> Point -> Point -> Point -> Prop;
conga_refl : forall A B C, ~ ColH A B C -> CongaH A B C A B C;
conga_comm : forall A B C, ~ ColH A B C -> CongaH A B C C B A;
congaH_permlr :
    forall A B C D E F, CongaH A B C D E F -> CongaH C B A F E D;
cong_5 :
    forall A B C A' B' C',
        ~ ColH A B C -> ~ ColH A' B' C' ->
        CongH A B A' B' -> CongH A C A' C' ->
        CongaH B A C B' A' C' ->
        CongaH A B C A' B' C';
same_side := fun A B l => exists P, cut l A P /\ cut l B P;
same_side' :=
    fun A B X Y =>
        X <> Y /\
        forall l, Incid X l -> Incid Y l -> same_side A B l;
outH :=
    fun P A B => BetH P A B \/ BetH P B A \/ (P <> A /\ A = B);
congaH_outH_congaH :
    forall A B C D E F A' C' D' F',
        CongaH A B C D E F ->
        outH B A A' -> outH B C C' -> outH E D D' -> outH E F F' ->
        CongaH A' B C' D' E F';
hcong_4_existence :
    forall A B C O X P,
            ~ ColH P O X -> ~ ColH A B C ->
            exists Y, CongaH A B C X O Y /\ same_side' P Y O X;
hcong_4_uniqueness :
    forall A B C O P X Y Y',
            ~ ColH P O X -> ~ ColH A B C ->
            CongaH A B C X O Y -> CongaH A B C X O Y' ->
            same_side' P Y O X -> same_side' P Y' O X ->
            outH O Y Y'
```

Fig. 6: Formalization of Group III, part 2: angle congruence axioms

```
Para := fun l m => ~ exists X, Incid X l /\ Incid X m;
euclid_uniqueness :
    forall l P m1 m2,
        ~ Incid P l ->
        Para l m1 -> Incid P m1-> Para l m2 -> Incid P m2 ->
        EqL m1 m2
```

Fig. 7: Formalization of Group IV

We denote by $\mathcal{V}$ the following set of axioms:

$$
\left\{A_{1}, A_{2}, A_{3}, A_{4}, A_{5}, A_{7}^{\prime}, A_{8}, A_{9}, A_{14}, A_{15}\right\}
$$

Figure 8 provides an overview of the structure of the proof.


Fig. 8: Overview of the proofs

### 4.1 Hilbert Plane is equivalent to Tarski A1-A9

The proof is divided into two parts.
First, we proved that the variant of Tarski $\mathcal{V}$ implies Tarski's axioms. The fact that the axiom $A_{6}$ can be derived from $\mathcal{V}$ is Theorem 2.17 of Gupta's Ph.D. Thesis [15].

Second, we proved that $\mathcal{V}$ follows from Hilbert's axioms. Hilbert's betweenness relation is strict, whereas Tarski's one is not. Obviously, we defined Tarski's betweenness relation (Bet) from Hilbert's one (BetH) as:

Definition Bet A B C : BetH A B C $\backslash / \mathrm{A}=\mathrm{B} \backslash / \mathrm{B}=\mathrm{C}$.
Hilbert's congruence relation is defined only for non degenerate segments, whereas Tarski's one include the case of the null segment:

```
Definition Cong A B C D :=
    (CongH A B C D }\\A<>B/\C<>D) \/ (A = B M C = D).
```

Axioms $A_{1}, A_{2}, A_{3}, A_{4}, A_{8}$ and $A_{14}$ are already axioms in Hilbert or easy consequences of the axioms. $A_{15}$ is a theorem in Hilbert which can be proved easily. Tarski's version of Pasch's axiom is stronger than Hilbert's one, because it provides information about the relative position of the points. We could recover the non degenerate case of Tarski's version of Pasch $\left(A_{7}^{\prime}\right)$ using some betweenness properties and repeated applications of Hilbert's version of Pasch. The five-segment axiom requires a longer proof. The non degenerate case is a trivial consequence of the Side-Side-Side theorems and the fact that if two angles are congruent their supplements are congruent as well. Those theorems are proved by Hilbert as theorems 18 and 14. To prove these two theorems we had to formalize the proof of Hilbert's theorems $12,15,16$ and 17 as well. Hilbert's proofs can be formalized without serious problem; we only had to introduce some lemmas about the relative position of two points and a line. For example, we had to prove that the same-side relation is transitive: if $A$ and $B$ are of the same side of $l$, and $B$ and $C$ are on the same side of $l$ then $A$ and $C$ are also on the same side of $l$. As already noticed by Meikle and Scott, these lemmas, which are as difficult to prove as Hilbert's other theorems are completely implicit in Hilbert's prose. The non-obvious part of the proof has been the degenerate case of the five-segment axiom and the upper 2-dimensional axiom $\left(A_{9}\right)$.

To prove the degenerate case of the five-segment axiom (when the point $D$ belongs to the line $A B$ ), we had to prove that then $D^{\prime}$ also belongs to line $A^{\prime} B^{\prime}$. We also had to prove many degenerate cases which reduce to segment addition and subtraction. Segment subtraction can be deduced from uniqueness of segment construction and from addition. We give here only the proof of the key lemma (Lemma 2 below), assuming the following lemma:

## Lemma 1.

$$
A \multimap B \multimap C \wedge A^{\prime} \nVdash B^{\prime} \hookleftarrow C^{\prime} \wedge A C \equiv_{H} A^{\prime} C^{\prime} \wedge A B \equiv_{H} A^{\prime} B^{\prime} \Rightarrow A^{\prime} \nsim B^{\prime} \nVdash C^{\prime}
$$

## Lemma 2.

$$
A \rightarrow B \rightarrow C \wedge A B \equiv_{H} A^{\prime} B^{\prime} \wedge B C \equiv_{H} B^{\prime} C^{\prime} \wedge A C \equiv_{H} A^{\prime} C^{\prime} \Rightarrow \mathrm{Col} A^{\prime} B^{\prime} C^{\prime}
$$

Proof. We will prove this lemma by contradiction so let us assume that $B^{\prime}$ does not belong to line $A^{\prime} C^{\prime}$. Let $B^{\prime \prime}$ be a point on $A^{\prime} C^{\prime}$ such that $A^{\prime} B^{\prime \prime} \equiv_{H} A B$. Let $C^{\prime \prime}$ a point such that $B^{\prime} C^{\prime \prime} \equiv_{H} B C$ and $A^{\prime} \nrightarrow B^{\prime} \nrightarrow C^{\prime \prime}$ (Fig. 9). So the triangle $B^{\prime} C^{\prime \prime} C^{\prime}$ is isosceles in $B^{\prime}$. Then Hilbert's theorem 12 lets us prove that $B^{\prime} C^{\prime} C^{\prime \prime} \widehat{=} B^{\prime} C^{\prime \prime} C^{\prime}$. By Lemma 1, we have that $A^{\prime}-B^{\prime \prime}-C^{\prime}$. We can derive $B^{\prime \prime} C^{\prime} \equiv_{H} B C$ by subtraction and then $A^{\prime} C^{\prime} \equiv_{H} A^{\prime} C^{\prime \prime}$ by addition. Therefore triangle $C^{\prime} A^{\prime} C^{\prime \prime}$ is isosceles in $A^{\prime}$, hence Hilbert's theorem 12 implies that $A^{\prime} C^{\prime \prime} C^{\prime} \widehat{=} A^{\prime} C^{\prime} C^{\prime \prime}$. By transitivity of angle congruence (Hilbert's theorem 19), we know that $B^{\prime} C^{\prime} C^{\prime \prime} \widehat{=} A^{\prime} C^{\prime} C^{\prime \prime}$. Finally we obtain a contradiction as the uniqueness of angle construction and the fact that $B^{\prime}$ and $C^{\prime \prime}$ are on the same side of $C^{\prime} C^{\prime \prime}$ let us prove that $C^{\prime} B^{\prime}$ is the same ray as $C^{\prime} A^{\prime}$.

The last axiom we need to prove is the upper 2-dimensional axiom. The proof is not completely straightforward because we do not assume decidability of intersection of lines: we can not distinguish cases to know if two lines intersect or not.

Let us first prove two useful lemmas.


Fig. 9: Proof of Lemma 2

Lemma 3. If two points $A$ and $B$ are not collinear with two points $X$ and $Y$, then either they are one the same side of the line $X Y$ or they are on opposite sides of this line.

Proof. First, one can construct a point $C$, such that points $A$ and $C$ are on the opposite side of the line $X Y$. Therefore, there exists a point $I$ collinear with $X$ and $Y$. If $A, B$ and $I$ are collinear, then either $A \multimap B \leftrightarrows I, B \multimap A \leadsto I$ or $A=B$ and then $A$ and $B$ are on the same side of line $X Y$ or $A \leadsto I \leadsto B$ and then $A$ and $B$ are on opposite sides of line $X Y$, as, neither $A$ nor $B$ can be equal to $I$ since they would then be collinear with $X$ and $Y$. Finally, if $A, B$ and $C$ are not collinear, then Pasch's axiom lets us conclude the proof.

In order to prove Lemma 5 we first prove a particular case which will be used repeatedly throughout this proof.

Lemma 4. If three distinct points $A, B$ and $C$ are equidistant from two different points $P$ and $Q$, then, assuming that $A$ is collinear with $P$ and $Q$, these points are collinear.

Proof. We know that neither $B$ or $C$ are collinear with $P$ and $Q$ because if they were then they would be equal to $A$, thus obtaining a contradiction. Therefore, using the previous lemma, either $B$ and $C$ are on opposite sides or on the same side of line $P Q$.

- If they are on opposite sides of line $P Q$, then we name $I$ the point of intersection between this line and the segment $\overline{B C}$. If we can prove that $A$ is equal to point $I$ we will be done. To do this we just have to prove that $I$ is equidistant from $P$ and $Q$. Using Hilbert's theorem 18 , we know that the angles $\widehat{P A B}$ and $\widehat{Q A B}$ are equal. Then Hilbert's theorem 12 lets us prove that $I$ is equidistant from $P$ and $Q$.
- If they are on the same side of line $P Q$, the previous lemma states that either $P$ and $Q$ are on opposite sides or on the same side of line $B C$.
- If they are on the same side, the uniqueness axioms let us prove that they are equal, therefore obtaining a contradiction.
- If they are on opposite side, then we name $I$ the point of intersection between this line and the line $P Q$. Without loss of generality, let us consider that $B$ is between $C$ and $I$ (if they are equal then $A, B$ and $C$ are trivially collinear). Using Hilbert's theorems 14 and 18, we know that the angles $\widehat{P B I}$ and $\widehat{Q B I}$ are equal. Then Hilbert's theorem 12 let us prove that $I$ is equidistant from $P$ and $Q$. Therefore $A$ is equal to $I$ and we are done.


## Lemma 5. If three distinct points $A, B$ and $C$ are equidistant from two different points

 $P$ and $Q$, then these points are collinear.Proof. We just have to consider the case where neither $A, B$ or $C$ are collinear with $P$ and $Q$ since otherwise the previous lemma lets us conclude. We know that either at least two of these points are on opposite sides of the line $P Q$ or all the points are one the same side of this line.

- If they are on opposite sides of line $P Q$, then we name $I$ the point of intersection between this line and the segment formed by the points which are on opposite sides of this line. As in the previous lemma we can prove that $I$ is equidistant from $P$ and $Q$. Then apply the previous lemma twice we know that $A, B$ and $I$ as well as $A, C$ and $I$ are collinear and the transitivity of collinearity allows us conclude.
- If they are on the same side of line $P Q$, either $P$ and $Q$ are on opposite sides or on the same side of line $A B$.
- If they are on the same side, the uniqueness axioms let us prove that they are equal, therefore obtaining a contradiction.
- If they are on opposite side, then we name $I$ the point of intersection between this line and the line $P Q$. As in the previous lemma we can prove that $I$ is equidistant from $P$ and $Q$. Then apply the previous lemma twice we know that $A, B$ and $I$ as well as $A, C$ and $I$ are collinear and the transitivity of collinearity allows us conclude.


### 4.2 Euclidean Hilbert Plane is equivalent to Tarski A1-A10

The fact that Playfair's postulate can be derived from Tarski's version of the postulate appears in Chapter 12 of [27], that we have formalized previously. The reverse implication and many other equivalence results have been described in [9]. For this implication, we have to assume the decidability of intersection of line: given two lines either they intersect or they do not.

## 5 Conclusion

We have obtained a formal proof that Hilbert's geometric axioms implies Tarski's ones. It is interesting to note, that to prove this result we had to correct the axioms we had given in [12], both removing unnecessary axioms and adding some properties. The connection between Hilbert's axioms and Tarski's ones allows us to obtain all the results we had proven in the context of Tarski's axiom system, in particular the culminating result which is the arithmetization of geometry.

In our experience, manipulating Hilbert's axioms is more tedious than Tarski's ones. As the axioms involve both points and lines (and planes in 3D), a single result can be stated in many different ways, and it requires formally a lot of administrative work to convert assumptions about lines into assumptions about points and vice-versa. Even if this administrative work can be partially automated, using for example the techniques proposed by Scott and Fleuriot [31], we have the impression that the mechanization of geometry based on a single primitive type as in Tarski's axiom system is more convenient. Moreover, Hilbert chose to exclude some degenerate cases from its axioms: angles are never null nor full, segments are never degenerate, this makes the proofs more tedious than in Tarski's setting. Still, Hilbert's axiom system has the advantage that axioms are grouped and each group can serve to develop a significant part of geometry, whereas in a development of geometry based on Tarski's axioms, one has to use both the betweenness and congruence axioms very early. Our formalization of Hilbert's axiom system avoids the concept of sets of points, segments, rays and angles. Our current formalization can hence be considered less faithful to the original than the one we presented in 2012, but this new formalization produces axioms which are both clearer and simpler to use.

## Perspectives

This work completes the formalization of a non trivial part of Hilbert's book. But, the formalization could be completed. Among the chapters which have not been formalized, we can cite the one about meta-mathematical results, and the one about arithmetization of hyperbolic geometry. It would also be useful to extend the formalization to 3D.

It would be also interesting to generate readable proofs from our Coq formalization to obtain a mechanized book about foundations of geometry.

We leave to future work the proof that our version of Hilbert's axiom is equivalent to a version in which angles are defined as a pair of rays.

## Availability

The full Coq development is available here (release 2.1.0):
http://geocoq.github.io/GeoCoq/

## References

1. Michael Beeson. Constructive geometry. In Proceedings of the Tenth Asian Logic Colloquium. World Scientific, 2010.
2. Michael Beeson. A constructive version of Tarski's geometry. Annals of Pure and Applied Logic, 166(11):1199-1273, 2015.
3. Michael Beeson and Larry Wos. OTTER proofs of theorems in Tarskian geometry. In Stéphane Demri, Deepak Kapur, and Christoph Weidenbach, editors, 7th International Joint Conference, IJCAR 2014, Held as Part of the Vienna Summer of Logic, Vienna, Austria, July 19-22, 2014, Proceedings, volume 8562 of Lecture Notes in Computer Science, pages 495-510. Springer, 2014.
4. Michael Beeson and Larry Wos. Finding Proofs in Tarskian Geometry. Journal of Automated Reasoning, submitted, 2015.
5. Michel Beeson. Proving Hilbert's axioms in Tarski geometry. http://www.michaelbeeson.com/research/papers/TarskiProvesHilbert.pdf, 2014.
6. Yves Bertot and Pierre Castéran. Interactive Theorem Proving and Program Development, Coq'Art: The Calculus of Inductive Constructions. Texts in Theoretical Computer Science. An EATCS Series. Springer, 2004
7. George D Birkhoff. A set of postulates for plane geometry, based on scale and protractor. Annals of Mathematics, 33:329-345, 1932.
8. Pierre Boutry, Gabriel Braun, and Julien Narboux. From Tarski to Descartes: Formalization of the Arithmetization of Euclidean Geometry. In The 7th International Symposium on Symbolic Computation in Software (SCSS 2016), EasyChair Proceedings in Computing, page 15, Tokyo, Japan, March 2016.
9. Pierre Boutry, Julien Narboux, and Pascal Schreck. Parallel postulates and decidability of intersection of lines: a mechanized study within Tarski's system of geometry. submitted, July 2015.
10. Pierre Boutry, Julien Narboux, and Pascal Schreck. A reflexive tactic for automated generation of proofs of incidence to an affine variety. October 2015.
11. Pierre Boutry, Julien Narboux, Pascal Schreck, and Gabriel Braun. Using small scale automation to improve both accessibility and readability of formal proofs in geometry. In Francisco Botana and Pedro Quaresma, editors, Preliminary Proceedings of the 10th International Workshop on Automated Deduction in Geometry (ADG 2014), 9-11 July 2014, University of Coimbra, Portugal, pages 1-19, Coimbra, Portugal, July 2014.
12. Gabriel Braun and Julien Narboux. From Tarski to Hilbert. In Tetsuo Ida and Jacques Fleuriot, editors, Post-proceedings of Automated Deduction in Geometry 2012, volume 7993 of $L N C S$, pages 89-109, Edinburgh, United Kingdom, September 2012. Springer.
13. Gabriel Braun and Julien Narboux. A synthetic proof of Pappus' theorem in Tarski's geometry. Journal of Automated Reasoning, 2015. accepted, revision pending.
14. Christophe Dehlinger, Jean-François Dufourd, and Pascal Schreck. Higher-Order Intuitionistic Formalization and Proofs in Hilbert's Elementary Geometry. In Automated Deduction in Geometry, volume 2061 of Lecture Notes in Computer Science, pages 306-324. Springer, 2001.
15. Haragauri Narayan Gupta. Contributions to the axiomatic foundations of geometry. PhD thesis, University of California, Berkley, 1965.
16. Robin Hartshorne. Geometry : Euclid and beyond. Undergraduate texts in mathematics. Springer, 2000.
17. David Hilbert. Foundations of Geometry (Grundlagen der Geometrie). Open Court, La Salle, Illinois, 1960. Second English edition, translated from the tenth German edition by Leo Unger. Original publication date, 1899.
18. David Hilbert. Les fondements de la géométrie. Dunod, Paris, jacques gabay edition, 1971 Edition critique avec introduction et compléments préparée par Paul Rossier.
19. Timothy James McKenzie Makarios. A Mechanical Verification of the Independence of Tarski's Euclidean Axiom. PhD thesis, Victoria University of Wellington, 2012. Master Thesis.
20. Laura Meikle and Jacques Fleuriot. Formalizing Hilbert's Grundlagen in Isabelle/Isar. In Theorem Proving in Higher Order Logics, volume 2758 of Lecture Notes in Computer Science, pages 319-334. Springer, 2003.
21. Richard S Millman and George D Parker. Geometry, A Metric Approach with Models Springer Science \& Business Media, 1991.
22. E.E. Moise. Elementary Geometry from an Advanced Standpoint. Addison-Wesley, 1990.
23. Julien Narboux. Mechanical Theorem Proving in Tarski's Geometry. In Francisco Botana Eugenio Roanes Lozano, editor, Post-proceedings of Automated Deduction in Geometry 2006, volume 4869 of Lecture Notes in Computer Science, pages 139-156, Pontevedra, Spain, 2008. Springer.
24. Moritz Pasch. Vorlesungen über neuere Geometrie. Springer, 1976.
25. William Richter. Formalizing Rigorous Hilbert Axiomatic Geometry Proofs in the Proof Assistant Hol Light.
26. Artur Rosenthal. Vereinfachungen des Hilbertschen Systems der Kongruenzaxiome. Mathematische Annalen, 71(2):257-274, June 1911.
27. Wolfram Schwabhäuser, Wanda Szmielew, and Alfred Tarski. Metamathematische Methoden in der Geometrie. Springer-Verlag, Berlin, 1983.
28. Phil Scott. Mechanising Hilbert's Foundations of Geometry in Isabelle. Master thesis, Edinburgh, 2008.
29. Phil Scott and Jacques Fleuriot. Compass-free Navigation of Mazes. In James H. Davenport and Fadoua Ghourabi, editors, SCSS 2016. 7th International Symposium on Symbolic Computation in Software Science, volume 39 of EPiC Series in Computing, pages 143-155. EasyChair, 2016.
30. Phil Scott and Jacques D. Fleuriot. An Investigation of Hilbert's Implicit Reasoning through Proof Discovery in Idle-Time. In Automated Deduction in Geometry - 8th International Workshop, ADG 2010, Munich, Germany, July 22-24, 2010, Revised Selected Papers, pages 182-200, 2010.
31. Phil Scott and Jacques D. Fleuriot. A Combinator Language for Theorem Discovery. In Intelligent Computer Mathematics - 11th International Conference, AISC 2012, 19th Symposium, Calculemus 2012, 5th International Workshop, DML 2012, 11th International Conference, MKM 2012, Systems and Projects, Held as Part of CICM 2012, Bremen, Germany, July 8-13, 2012. Proceedings, pages 371-385, 2012.
32. Alfred Tarski and Steven Givant. Tarski's System of Geometry. The bulletin of Symbolic Logic, 5(2):175-214, June 1999.

[^0]:    ${ }^{1}$ We use the numbering of theorems as of the tenth edition.

